/*
Jonathan Hill	
Proposal title: Giving as Habit? Exploring Modes of Cognition and Charitable Donations


HYPOTHESES 

Stated-Hyp1: H1: The effects of previous giving habits on donating will be the greatest in the cognitive load prime when System 2 deliberation is blocked (p. 5)

	Test-Hyp1: The effect of previous giving on current giving will be higher in those in the cognitive load condition vs. control. [previous giving measured as whether any donation made in the past]

	Test-Hyp2: The effect of previous giving on current giving will be higher in those in the cognitive load condition vs. control. [previous giving measured as frequency of past donation]
	
	
Stated-Hyp2: H2: The effects of previous giving habits on donating will be the least when System 2
deliberation is primed through mathematical word problems involving money (p. 5)

	
	Test-Hyp3: The effect of previous giving on current giving will be higher in those in the cognitive load condition vs. deliberation condition.	[previous giving measured as whether any donation made in the past]

	Test-Hyp4: The effect of previous giving on current giving will be higher in those in the cognitive load condition vs. deliberation condition. [previous giving measured as frequency of past donation]
	
********************************************************************************
NOTES
- There are several questions about previous giving, however, some of those seem 
less pertinent. We are not including the following:

- whether gave to religious orgs specifically or non-religious orgs specifically. 
[Ignoring this because Q5 already captures this and type of previous giving is 
not relevant to the hypothesis.]	
- amount of money donated [Ignoring this because this will depend on means]
	
- Outcome: There are 2 possible outcomes: whether R wants to donate at all and 
amount of money donated. The participants are shown the amount question only if 
they want to donate at all. We are only including the first question as the 
outcome because the second question applies to a self-selected sample.	
*/

clear all
use "Hill329.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

	tab XTESS126
	recode XTESS126 (3=0 "0 control") (1=2 "2 cognitive load") (2=1 "1 economic deliberation"), gen(treatment)
	tab treatment

	
* OUTCOME MEASURES
	/*
	Q2a.  We are asking all of our respondents if they would like to donate any of their Bonus Points to the Special Olympics. The Special Olympics is a nonprofit organization that provides year-round sports training and athletic competitions for those with intellectual disabilities. Would you like to donate? 
	*/
	tab Q2a
	recode Q2a (1=1) (2=0), gen(donates)
	tab donates
 	
* MODERATORS

* previous giving
	* see notes

	// Q5.	During the year 2012, did you, or anyone in your family, donate household money, assets, or property/goods to religious or charitable organizations?
	tab Q5
	recode Q5 (-1=.) (1=1) (2=0), gen(past_donated)
	
	// Q9 How frequently, during 2012, did you or your family donate money during a worship service? 
	tab Q9
	replace Q9=. if Q9==-1
	clonevar past_donatefrequency = Q9
	
********************************************************************************

* ANALYSIS
	
*	Test-Hyp1: The effect of previous giving on current giving will be higher in those in the cognitive load condition vs. control. [previous giving measured as whether any donation made in the past]

	reg donates i.treatment##i.past_donated if treatment!=1
	// reject. 0.930
	tess 2.treatment#1.past_donated +, init(Hill329)
	
*	Test-Hyp2: The effect of previous giving on current giving will be higher in those in the cognitive load condition vs. control. [previous giving measured as frequency of past donation]
	
	reg donates i.treatment##c.past_donatefrequency if treatment!=1
	// reject. 0.179	
	tess 2.treatment#c.past_donatefrequency +
	
*	Test-Hyp3: The effect of previous giving on current giving will be higher in those in the cognitive load condition vs. deliberation condition.	[previous giving measured as whether any donation made in the past]

	reg donates i.treatment##i.past_donated if treatment!=0
	// reject. 0.548
	tess 2.treatment#1.past_donated +

*	Test-Hyp4: The effect of previous giving on current giving will be higher in those in the cognitive load condition vs. deliberation condition. [previous giving measured as frequency of past donation]
	
	reg donates i.treatment##c.past_donatefrequency if treatment!=0
	// reject. 0.734	
	tess 2.treatment#c.past_donatefrequency +	
